A sample solution of Problem 2 in Problem Set 19 with `Macaulay2`

Chris Peterson and Hirotachi Abo

Partial differentiations.

Let k be an algebraically closed field, and let R denote the polynomial ring $k[x, y, z]$. Assume that I is an ideal with $V(I) = \emptyset$. Then the quotient ring Q of R modulo I is a finite-dimensional vector space over k. In other words, Q is artinian. In this case, Q has a free resolution of the following type (this is not trivial):

$$0 \to F_3 \to F_2 \to F_1 \to R \to Q \to 0,$$

where F_i are free modules for each $i = 1, 2$ and 3. This ring is called *arithmetically Gorenstein* if F_3 can be written as $R(-l)$ for some positive integer l (i.e. F_3 has rank 1).

Remark 1. Let Q be an artinian arithmetically Gorenstein ring over R. Then there is a positive integer d such that $\dim_k(Q_d) = 1$ and $\dim_k(Q_i) = 0$ for all $i > d$.

Let S be the polynomial ring $k[X, Y, Z]$, and let S act on R by partial differentiation:

$$X(x) := \partial_x(x), Y(y) := \partial_y(y) \text{ and } Z(z) := \partial_z(z).$$

Let F be a single homogeneous polynomial of degree d in R. For this F, denote by $I_S(F)$ the set of polynomials G in S satisfying $G(F) = 0$. Then $I_S(F)$ is an ideal of S (why?). Consider the quotient ring $Q(F)$ of S modulo $I_S(F)$. It is known that this ring is artinian and arithmetically Gorenstein.

Remark 2. By definition, $\dim_k(Q(F))_i = 0$ for all $i > d$. It immediately follows that $Q(F)$ is artinian. Let G be a degree d homogeneous polynomial in $I_S(F)$:

$$G = \sum_{i+j+k=d} a_{ijk} x^i y^j z^k.$$

Then $G(F)$ can be written as a k-linear combination of a_{ijk}'s. So the elements G satisfying $G(F) = 0$ form a one-codimensional subspace in S_d, and hence $\dim_k(Q(F)_d) = 1$. In general, we have the following equations:

$$\dim_k(Q(F)_r) = \dim_k(Q(F)_{d-r}) \text{ for all } 0 \le r < d/2.$$

Proposition. If Q is an artinian arithmetically Gorenstein ring of S, then there is a polynomial F in R such that $Q = S/I_S(F)$. Furthermore, such a polynomial is uniquely determined up to constants.

Proof. See *Inverse System of a Symbolic Power I* in Journal of Algebra **174**, 1080-1090, by J. Emsalem and A. Iarrobino. □

Let us discuss how to compute the corresponding polynomial F in R from a given artinian arithmetically Gorenstein ring Q of S. From Remark 1, it follows that there is a positive integer d such that $\dim_k(Q_d) = 1$ and $\dim_k(Q_i) = 0$ for $i > d$. Let I be the ideal in S, that is obtained as the kernel of the ring homomorphism from S to Q, and let $\{f_1, \ldots, f_t\}$ be a set of generators of I_d, where

$$t = \dim_k(S_d) - \dim_k(Q_d) = \binom{d+2}{2} - 1.$$

Consider the bilinear map $\tilde{T}$ from $I_d \otimes_k R_d$ to k defined by $\tilde{T}(G \otimes F) = G(F)$. Recall that this bilinear map corresponds to a linear transformation T from R_d to $(I_d)^*$. The nullspace of this linear transformation, that is equal to the set

$$\mathfrak{F} = \{F \in R_d \mid G(F) = 0 \text{ for all } G \in I_d\},$$

has dimension 1. Let F be a nonzero polynomial in $\mathfrak{F}$. Such a polynomial can be computed explicitly by using the matrix representation of T with respect to the basis $\{f_1^*, \ldots, f_t^*\}$ for $(I_d)^*$ and the standard basis for R_d. Indeed, this matrix is given by $\begin{pmatrix} f_1 & \cdots & f_t \end{pmatrix}^T \cdot \begin{pmatrix} x_0^d & \cdots & x_2^d \end{pmatrix}$. Here is an algorithm for finding F:

```
Input: ideal I with Q=S/I artinian, arithmetically Gorenstein
Output: a nonzero polynomial F with I_S(F)=I
i:=0
r:=dim(Q_0)
d:=0
Repeat
   r=dim(Q_i)
   d=i-1
Until r=0
B:=a basis of I_d
B':=the standard basis for R_d
A:=B^T*B'
syz:=a syzygy matrix of A
F:=B'*syz
```

In `Macaulay2`, we use the function `diff` to compute `A` in pseudocode. This function is used to differentiate polynomials. Basically, the first argument is the variable to differentiate with respect to, and the second one is the polynomial to be differentiated:

```
i1 : R=QQ[x,y]

o1 = R

o1 : PolynomialRing

i2 : F=x^2*y+y^7

      7    2
o2 = y  + x y

o2 : R

i3 : diff(x,F)

o3 = 2x*y

o3 : R
```

The first argument can be also sum:

```
i4 : diff(x+y,F)

       6    2
o4 = 7y  + x  + 2x*y

o4 : R
```

The first and second arguments can be matrices:

```
i5 : diff(transpose matrix{{x,y}},matrix{{x^3+y,x*y+y^2}})

o5 = {1} | 3x2 y    |
     {1} | 1   x+2y |

                 2       2
o5 : Matrix R  <--- R
```

This corresponds to the jacobian matrix of the ideal generated by the matrix in the second argument.

Here is the function for finding F:

```
i6 : idealOfCurveCorrToGorenstein=(idl)->(
          i:=0;
          isMaximum:=false;
          r:=ring idl;
          numbasis:=numgens source basis(0,r/idl);
          maxi:=0;
          while not isMaximum do (
          numbasis=numgens source basis(i+1,r/idl);
          maxi=i;
          if numbasis===0 then (
               isMaximum=true;
               g:=(gens idl)* map(source gens idl,basis(maxi,idl));
               m:=basis(maxi,r);
               mat:=diff(transpose g,m);
               sy:=syz mat;
               f:=basis(maxi,r)*sy;
               );
          i=i+1;
          );
        ideal f)

o6 = idealOfCurveCorrToGorenstein

o6 : Function
```

Problem 2 (Set 19). Let $J = (6xz - 5z^2, 6y^2 - 4z^2, 6xz - 3z^2, 6xy - 2z^2, 6x^2 - z^2)$. Then the quotient ring Q of S modulo J is artinian and arithmetically Gorenstein. To check this, compute the free resolution of Q:

```
i7 : KK=QQ;

i8 : ringP2=KK[x,y,z];

i9 : J=ideal(6*y*z-5*z^2,6*y^2-4*z^2,6*x*z-3*z^2,6*x*y-2*z^2,6*x^2-z^2);

o9 : Ideal of ringP2

i10 : fJ=res J;

i11 : betti fJ
```

```
o11 = total: 1 5 5 1
          0: 1 . . .
          1: . 5 5 .
          2: . . . 1
```

The free resolution of Q is of length 4, and its last spot has rank 1. So Q is an artinian and arithmetically Gorenstein ring. By using the function `idealOfCurveCorrToGorenstein`, we can compute the degree 2 polynomial F in R such that $J = I_S(F)$:

```
i12 : F=idealOfCurveCorrToGorenstein(J)

            1  2   2       2  2         5        2
o12 = ideal(-*x  + -*x*y + -*y  + x*z + -*y*z + z )
            6      3       3            3

o12 : Ideal of ringP2
```